
Families of Random Variables
J. C. Hudson

Name, parameters, f(x), support, E(X) and Var(X) are listed. F(x) is listed below f(x) if there is a useful form. See 
ks.doc for references.

,   sgn(x) = U(x)-U(-x),  

 is the set of integers,  + is the set of positive integers, ℝ is the set of reals.
Γ(x) is the gamma function.

Name    Parameters  Density/CDF Support E(X),Var(X)

Arcsin a>0 -a<x<a 0, a²/2

Bernoulli 0<p<1 {0,1} p, p(1-p)

Beta 0<x<1 , 

Binomial np, np(1-p)

Cauchy , 

, 

Chi-Square x>0 ν, 2ν

x>0 , 



Erlang x>0 nθ, nθ2

 

Exponential   μ>0 x>0 μ, μ2 

F   x>0

Noncentral F  x>0      E(X) = , ν2 > 2
                             Var(X) = , ν2 > 4

Gamma x>0 aθ, aθ2

Geometric 0<p<1

   

x>0 μ, 



Name    Parameters Density/CDF Support E(X),Var(X)

Laplace    

Logistic μ, σ2

Lognormal   x>0

 μ, σ2

0, 1

Pareto b>0 x>1

    

Poisson μ>0 μ, μ

Rayleigh b>0 x>0

Student's t  

x∈ℝ



Name    Parameters Density/CDF Support E(X),Var(X)

Triangular  a-b<x<a+b

 a<b a<x<b

0<x<1

Weibull x>δ



Probability Program for Small Computers

This program is written in Sharp EL-5500III BASIC. Extra spaces are used for clarity.
sqr is used instead of Sharp's √ and pi is used instead of π. Expect to make 
substantial changes in i/o statements if you use the program on another machine.
       
       lines       compute or approximate
       
       100 - 150   binomial pmf and cdf
       200 - 290   hypergeometric pmf and cdf
       300 - 350   Poisson pmf and cdf
       400 - 470   Gaussian right tail probability
       500 - 550   Student's t right tail probability
       600 - 630   chi - square right tail probability
       700 - 750   f distribution right tail probability 
100 input "x=";x,"n=";n,"p=";p
110 t = (1-p)^n : s = t
120 if x = 0 then 140
130 for i = 1 to x : t = t*(n-i+1)*p/i/(1-p) : s=s+t : next i
140 using : print x;" ";n;" ";p
150 print using "###.####";t+0.00005;s+0.00005 : goto 100

200 input "x=";x,"np=";np,"ns=";ns,"k=";k
210 mi = 0 : if k+ns-np > 0 then let mi=k+ns-np
220 ma = k : if ns < k then let ma = ns
230 if x < mi then let t = 0 : let s = 0 : goto 280
240 if x > ma then let t = 0 : let s = 1 : goto 280
250 j=np-ns : d=k : nt=ns : if mi > 0 then let j=ns : let d=np-k : let nt=np-ns
260 t=1 : for i=1 to nt : j=j+1 : t = t*(j-d)/j : next i : s=t : if x=mi then 280
270 for i = mi+1 to x : t = t*(k-i+1)*(ns-i+1)/i/(np-k-ns+i) : s = s+t : next i
280 using : print x;" ";np;" ";ns;" ";k
290 print using "###.####";t+0.00005;s+0.00005 : goto 200

300 input "x=";x,"m=";m
310 t = exp(-m) : s = t
320 if x = 0 then 340
330 for i = 1 to x : t = t*m/i : s = s+t : next i
340 using : print x;" ";m
350 print using "###.####";t + 0.00005;s + 0.00005 : goto 300

400 input "z=";z : gosub 420
410 using : print z;" qz=; using "##.####";q + 0.00005 : goto 400

420 y = exp(-z*z/2) / sqr(2*pi)
430 w = 1 / (1 + 0.2316419 * abs(z))
440 q=0.31938153+w*(-0.356563782+w*(1.781477937+w*(-1.821255978+w*1.33027449)))
450 q = q*y*w : if z < 0 then let q = 1-q
460 return

500 input "df=";d,"t=";t : if d < 2 then 500



510 g = sqr(ln(1+t*t/d))
520 s = 0.184 * (8*d+3)/d/g
530 z = sqr(d) * g * (1 - 2*sqr(1 - exp(-s*s)) / (8*d+3))
540 gosub 420 : if t < 0 then let q = 1-q
550 using : print d;t;" qt=";using "###.####";q + 0.00005 : goto 500

600 input "df=";d,"x=";x:if d < 3 then 600
610 z = sqr(x + (d - 1) * (ln((d - 1) / x) - 1))
620 z = z * (x - d + 2/3 - 0.08 / d) / abs(x - d + 1) : gosub 420 
630 using : print d;x;"qx=";using "###.####";q + 0.00005 : goto 600

700 input "df1=";d,"df2=";e,"f=";f
710 p = e / (d * f + e) : c = 0.08 * ((1-p) / e - p/d + (.5-p)/(d+e))
720 d = d-1 : e = e-1 : z = e * ln(e/((d+e)*p)) + d * ln(d/((d+e)*(1-p)))
730 z = sqr(3 * (d + e) * z / (3 * (d + e) + 1))
740 z = z * (e+1/3-(d+e+2/3)*p+c) / abs(e - (d+e)*p) : gosub 420 
750 using : print d+1;e+1;f;"qf=";using "###.####";q + 0.00005 : goto 700

Thisted [1988] gives the Peizer and Pratt [1968] approximation for the chi-square 
and F distributions and the Wallace [1959] approximation for the t distribution. The
Gaussian right tail approximation is given by McConnell [1990].

Errors
       
The binomial, Poisson and hypergeometric computations are exact to 4 decimal 
places for all values checked. The program does very well approximating Normal 
upper tail probabilities. For Student's t, upper tail probabilitiess are acceptable for 
4 or more df, and are acceptable for 8 or more df for chi-square. for the F 
distribution, results are ok if both df are 8 or more, results are not acceptable if 
both are 4 or less.
       
The program crashes if t = 0, x² = df - 1 or f = 1 with both degrees of freedom 
equal. The program will not accept df = 1 for Student's t or df < 3 for x².
       


